# Maintainer: Caleb Maclennan <caleb@alerque.com>
# Maintainer: Morgan <morganamilo@archlinux.org>
# Contributor: Dave Reisner <d@falconindy.com>

pkgname=expac
pkgver=10
pkgrel=5
pkgdesc='alpm data (pacman database) extraction utility'
arch=(loong64 x86_64)
url="https://github.com/falconindy/$pkgname"
license=(GPL)
depends=(pacman)
makedepends=(git
             meson
             perl)
source=("git+$url.git?signed&tag=$pkgver")
validpgpkeys=('487EACC08557AD082088DABA1EB2638FF56C0C53') # Dave Reisner
md5sums=('SKIP')

build() {
	arch-meson "$pkgname" build
	meson compile -C build
}

package() {
	meson install -C build --destdir "$pkgdir"
}
